

Some Uses of Roles in Blaise 5

G J Boris Allan, Westat

A field can be presented in many different ways, and in Blaise 4.8 any field might have different standard ways in which it might be presented. That is, a field always had a field name, it might have a tag, it might have question text in one or more languages, and it might have a label or description. Datamodels would have titles, categories usually have labels, and the list continues.

In Blaise 5, there are six predefined text roles, and the user can extend this list to include as many as needed for the instrument being designed. Here are some text roles defined for one survey (and we might still extend the list):

```
ROLES =  
  HelpInt, HelpResp,  
  Watermark, Units, Template, ToolTip, EditMask, Width, QText,  
  PreInst, SAQ, QNum, InstInt, InstResp,  
  NaText, SaVisible, InitYear, InitMonth, InitDay
```

In the following we discuss three distinct examples:

- a simple extension of texts to present different information to users of disconnected (interviewer laptop) interviews and users of web browsers (self-administered questionnaires);
- extending the friendliness of numerical text boxes;
- using role texts to give an easily accessible way of collecting different types of information for comment review.

1. Interviews and Self-administered Questionnaires

We have a study that wants to start interviewing with an interviewer (an interview), and then possibly continue data collection with the respondent connecting to the survey by means of a web browser or a mobile device (an SAQ – self-administered questionnaire). The text for the questionnaire will, for many questions, be the same for the interview and for the SAQ, but there will be differences for some questions. It was decided that, for a question, there were certain functional areas:


- Interviewer instructions prior to the respondent question (PreInst) not appropriate for an SAQ.
- A question asked of the respondent in an interview or an SAQ (the default).
- A question only asked in the case of an SAQ, if the default text did not make sense for a web question (SAQ).
- Interviewer instructions to the respondent on how to complete the answer (InstInt).
- SAQ instructions to the respondent on how to complete the answer (InstResp).

One other role (QNum) gives the question number displayed in case of problems – the tag is used to hold the SAS variable label, if applicable.

Here is the Blaise 5.4 code:

```
CoolRoof
  "Please look at Show Card A4. Does the roof of this building have any of the following
characteristics that
  allow it to reflect more sunlight or absorb less heat than a standard roof?"
  PreInst "SHOW CARD A4"
  InstInt "Enter all that apply"
  SAQ "Does the roof of this building have any of the following characteristics
that allow it to reflect more sunlight or absorb less heat than a standard roof?"
  InstResp "Please select all that apply"
  QNum "A11"
/ "Cool roof materials"
: SET OF
  (CoatingPaint "White or highly reflective coating or paint",
  TilesShingles "White or highly reflective tiles or shingles",
  Aluminum "Aluminum coating",
  Ballasted "Ballasted roof system",
  Vegetative "Vegetated roof",
  Other "Other",
  None "None of these")
```

Here is the screen produced for the interview question:



SHOW CARD A4

Please look at Show Card A4. Does the roof of this building have any of the following characteristics that allow it to reflect more sunlight or absorb less heat than a standard roof?

Enter all that apply

- White or highly reflective coating or paint
- White or highly reflective tiles or shingles
- Aluminum coating
- Ballasted roof system
- Vegetated roof
- Other
- None of these

Next

The same question in SAQ form is:

Does the roof of this building have any of the following characteristics that allow it to reflect more sunlit
Please select all that apply

White or highly reflective coating or paint

White or highly reflective tiles or shingles

Aluminum coating

Ballasted roof system


Vegetated roof

Other

None of these

Next

Let us take a look at the PreText control in Vertical Field Pane template in the resource database (the blrd file):




We can see that for the pretext control both the RoleName and the ScreenReaderRoleName properties are equal to the PreInst role, What controls whether we see the pretext control is its Visibility as defined by an expression:

```
IF LEN(Field.GetRoleText('PreInst')) = 0 OR State.IsCawi THEN  
    'Collapsed'  
ELSE  
    'Visible'  
ENDIF
```

That is, IF there is no role text for PreInst (the length of the text is zero) OR the the questionnaire is being viewed in a browser (IsCawi) THEN the control is collapsed, ELSE the role text is visible. All the other

field texts have similar expressions to control visibility. Note that the font format for PreInst has a different colour text by default, partly so that interviewer instructions are clearer to the interviewer.

Looking at the two controls InstInt and InstResp we can see that they are the reverse of each other (that is, both cannot be visible at the same time), and both have a different font format from standard question text – the font format is (like PreInst) defined in the resource database:



The expression for the visibility of InstResp is:

```
IF LEN(Field.GetRoleText('InstResp')) = 0 OR (NOT State.IsCawi) THEN
  'Collapsed'
ELSE
  'Visible'
ENDIF
```

2. Enhancing Data-collection Controls

Take a look at this question:

What is the gross or total square footage of all the space in this building shafts, and indoor parking levels?

square feet

I don't know


Next

There are three points to note: there is a watermark (a standard Blaise 5 feature); there is a description of the units for the number following the entry box; and there is an explicit DK with the entry box (this question usually hides DK/RF when a question is first asked).


Here is the question code (note that fields in this instrument have DK and RF by default, which is why this field has NORF to switch off the refusal option):

```
SqFt (A6) "What is the gross or total square footage of all the space in this building
both finished and unfinished, including basements, hallways, lobbies,
stairways, elevator shafts, and indoor parking levels?"
Watermark "Enter a number"
Units "square feet"
SaVisible "Yes"
/ "Square feet"
: 1..99999999, NORF
```

A label (numberUnits) has been added to the template for a NumberTextBox, and the label text is taken from the Units text role – if the Units role does not exist for a question, the label is empty:




The appearance of the DK is controlled by the SaVisible text role, which is applied at run time to the RadioButtonOptionalHidden template (remember NORF):



3. Annotating interviews in Blaise 5

Take this page with two questions:




Please enter the date this occurred, using the date-picker calendar.

11/29/2017

Please enter the time this occurred.

1:00 PM

We press F9 (the default for remarks is Ctrl-M, but we added F9):¹



Please enter the date this occurred, using the date-picker calendar.

11/29/2017

Please enter the time this occurred.

1:00 PM

Information_Source

Observation Respondent OtherFamily

Information

That is, before we enter some information, we are asked to provide the source of the information.²

¹ Thanks to Ralph Dohlmans of Statistics Netherlands for his creativity in helping me explore this topic.

² Thanks to Peter Stegehuis of Westat for his work on sophisticated remarks in Blaise 4.8.

Later on, we have:

Which of the following is true for you?

Select one option only.

- One
- Two
- Three
- Four
- Five
- Six
- Seven
- Eight
- Nine

After F9:

Which of the following is true for you?

Select one option only.

- One
- Two
- Three
- Four
- Five
- Six
- Seven
- Eight
- Nine

Comment Importance

Low Medium High


Comment

Before we comment we are asked how important is that comment.

A final example:

Open-ended question

After F9:



The screenshot shows a light blue background with a grey rectangular area containing two input fields. The top field is labeled 'Open-ended question' and is empty. Below it is a field labeled 'Information', also empty. The labels are in a light blue font.

This time we are simply asked for the Information, but not asked about the source.

For many other fields F9 has no effect.

An important introduction in Blaise 5 is the notion of field properties, and the beginning of this questionnaire code includes two significant sections – ROLES and FIELDPROPERTIES:


```
...
ROLES = HELP, ShowFieldProperty
...
FIELDPROPERTIES
  Comment_Importance: (Low, Medium, High)
  Comment: STRING[1023]
  Information_Source: (Observation, Respondent, OtherFamily)
  Information: STRING[1023]
...
```

In the FIELDPROPERTIES section there are definitions of the four types of remark we have encountered above and, to define which combination of remarks we show, we use the ShowFieldProperty role. For example, here is the definition of the time question:

```
TimeField ENG "Please enter the time this occurred."
  ESP "Introduce la fecha esto seleccionando la hora."
  ShowFieldProperty "Information_Source;Information;"
  / "Example time field"
  : TIMETYPE, DK
```

The ShowFieldProperty role specifies "Information_Source;Information;", that is, show the information entry field and ask for the source of the information. If there is no ShowFieldProperty then F9 will not

have any effect on the display. So how do the field properties get shown? We go to the Field Property templates, and modify visibility for a field property listed in the ShowFieldProperty role text:



If a field property is not listed it is not shown.

It is of note that there is more to field properties than mere remarks. For example, we can filter on the information source:

```

IF (TimeField.Information_Source = Observation) THEN
  aString := TimeField.Information
ENDIF

```

And so forth.